# $Id$
# Maintainer: Maxime Gauduin <alucryd@archlinux.org>
# Contributor: Tom Gundersen <teg@jklm.no>
# Contributor: John Proctor <jproctor@prium.net>

pkgname=libogg
pkgver=1.3.3
pkgrel=3
pkgdesc='Ogg bitstream and framing library'
arch=('x86_64')
url='http://www.xiph.org/ogg/'
license=('BSD')
depends=('glibc')
source=("https://downloads.xiph.org/releases/ogg/libogg-${pkgver}.tar.xz")
sha256sums=('4f3fc6178a533d392064f14776b23c397ed4b9f48f5de297aba73b643f955c08')

build() {
  cd libogg-${pkgver}

  ./configure \
    --prefix='/usr'
  make
}

package() {
  cd libogg-${pkgver}

  make DESTDIR="${pkgdir}" install
  install -Dm 644 COPYING -t "${pkgdir}"/usr/share/licenses/libogg/
}

# vim: ts=2 sw=2 et:
